

PharmaSUG2011 - Paper SP06

Caution: Hazards Crossing! Using the Renyi Test Statistic in Survival Analysis

Matthew Davis, M.S., University of Pennsylvania, Philadelphia, PA
Sharon X. Xie, Ph.D., University of Pennsylvania, Philadelphia, PA

1. ABSTRACT

One of the main assumptions regarding a number of survival analysis techniques is that of proportional hazards. If the hazard rates are found to cross, analyses such as Cox proportional hazards regression and log-rank test will either lose power or be rendered inadequate. One way to adequately analyze survival data with crossing hazard rates is to obtain inference using the Renyi test statistic. Both an example and a macro will be provided to calculate the Renyi test statistic with various weights.

2. INTRODUCTION

Survival analysis is a breakthrough in analyzing statistics that has become invaluable to the analysis of pharmaceutical data. It uses all available information of the time to event data by appropriately incorporating censoring information. It benefits greatly from the Martingale counting process instead of any specific distribution. For comparison of survival curves among groups, weighted or unweighted log-rank tests are often used. These tests were based on the weighted integral of estimated difference between the cumulative hazard rates. For two sample tests, the log-rank test has optimal local power to detect differences in the hazard rates, when the hazard rates are proportional (Klein 2003). However, when the hazard rates cross, these tests have little power. Similarly, Cox proportional hazard models require the hazards are proportional.

This paper will focus primarily on survival analysis techniques on group comparisons when the hazard rates cross (non-proportional hazards). When we find evidence of non-proportional hazards, the analysis of the data using classical log-rank or Cox Proportional Hazards can become more difficult. While there are certainly ways to analyze the data in spite of this weakness, particularly for Cox Proportional Hazards models, it would be helpful to have a resource that allows us to analyze the data in the most parsimonious manner. In that light, the Renyi family of test statistics can be computed which, similar to the Log-Rank or Wilcoxon tests in Kaplan Meier analysis, will help determine if there is a significant difference in survival rate between two groups given the presence of crossing hazard rates. To this point, SAS has not incorporated the Renyi family of statistics into their PROC LIFETEST procedures. Therefore, after a brief history of the Renyi family of statistics and some of the math behind these analyses, an example will be given where the Renyi statistic is appropriate along with the SAS code needed to perform the analysis.

3. HISTORY OF RENYI STATISTICS

The idea of the Renyi statistics started with Dr. Alfred Renyi in his text *On the Theory of Order Statistics* (Renyi, 1953). It is in this work that he used the ideas of counting processes and Standard Brownian Motion to apply to survival analysis in a similar way the Kolmogorov and Smirnov had years prior (see a nice description in Schumaker, 1984). It wasn't until a number of years later, in 1980, when R.D. Gill in his work *Censoring and Stochastic Integrals* brought life to this statistic and clearly defined the use for general survival analysis specific to situations with crossing hazard functions. At this point, the Renyi family of statistics was catapulted from the annals of journals into applied mathematics. These statistics caught the eye of Martin Schumacher in his article "Two-Sample Tests of Cramer-von-Mises and Kolmogorov-Smirnov Type for Randomly Censored Data". While this paper originally set out to test the Cramer-von-Mises and Kolmogorov-Smirnov tests against the more widely used Log-rank (Peto & Peto, 1972) and Wilcoxon (Gehan, 1965) tests, the Renyi family of test statistics were also compared with these methods documenting in the literature that the Renyi family of statistics delivered comparable results to the Log-rank and Wilcoxon tests. From this point, being validated as an appropriate (and more conservative) method for analysis of survival data, various texts such as *Counting Processes and Survival Analysis* (Fleming and Harrington, 1991) and *Survival Analysis: Techniques for Censored and Truncated Data* (Klein and Moeschberger, 2003) mentioned this family of statistics as alternatives to classical analysis.

4. THE MATHEMATICS BEHIND RENYI STATISTICS

4.1 Kaplan-Meier Estimates and Log-Rank Tests

The Renyi family of statistics combine the non-parametric Kaplan-Meier analysis with Standard Brownian motion to allow for inference to be made on cases in survival analysis where the hazard ratios cross. Consider a collection of data points $(X_{11}, C_{11}), (X_{12}, C_{12}) \dots (X_{1N}, C_{1N}), \dots, (X_{k1}, C_{k1}), (X_{k2}, C_{k2}) \dots (X_{kN}, C_{kN})$ where X is the survival time and C is the right censoring time from the i^{th} individual in the k^{th} group. Associated with each subject (k,i) will be the indicator variable $\Delta_{ij} = 1_{(X < C)}$ which equals one if the event occurred and zero otherwise. Y_k denotes the number of people at risk in group k . The survival time for each group k at time t can be represented as

$$\widehat{S}_k(t) = \prod_{i: X_{K(i)} < t} \left(\frac{Y_k(X_{K(i)}) - 1}{Y_k(X_{K(i)})} \right)^{\Delta_{k(i)}}$$

while the cumulative hazard function can be estimated as

$$\widehat{V}_k(t) = \sum_{i: X_{K(i)} < t} \frac{\Delta_{k(i)}}{Y_k(X_{K(i)})}$$

These values provide the point estimates of survival and cumulative hazard at each time point. In order to calculate the normal log-rank statistic, we can calculate the statistic

$$Z_k(t) = \sum_{j=1}^{\Sigma \Delta_i} W(t_i) \left[\Delta_{ji} - Y_{ji} \left(\frac{\Delta_i}{Y_j} \right) \right], j = 1 \dots K$$

where W is a weight function (which will be described later). The variance of this estimate is

$$\widehat{\sigma}_{kk} = \sum_{j=1}^{\Sigma \Delta_i} W(t_i)^2 \frac{Y_{ji}}{Y_j} \left(1 - \frac{Y_{ji}}{Y_j} \right) \left(\frac{Y_j - \Delta_j}{Y_j - 1} \right), j = 1 \dots K$$

When calculating the classical log-rank or Gehan (Wilcoxon) statistic, the test-statistic is simply found using the formula $Z = \frac{Z_k(t)}{\sqrt{\widehat{\sigma}_{kk}}}$ which follows a standard normal distribution for large samples (Klein, 2003).

4.2 Weight Functions

Many different weight functions have been applied to the Log-rank tests, most of which to emphasize different aspects of the survival curve. As will be shown, these weights also apply to the Renyi family of statistics. The following is a list of weight functions commonly used:

1. Log-Rank: $W(t) = 1$. This weight is the most general and applies equal weight to all parts of the survival curve. The results from this test are also comparable to the results of a Cox Proportion Hazards model, allowing for the same inference to be made by different methods.
2. Gehan (Wilcoxon): $W(t) = Y_i$. This weight emphasizes early differences in the survival curve as each statistic is weighted by the at risk set at a given time point.
3. Tarone-Ware: $W(t) = Y_i^{1/2}$. According to Klein (2003), "This class of weights gives more weight to differences between the observed and expected number of deaths in sample j at time points where there is the most data".
4. Peto-Peto: $W(t) = \bar{S}(t_i)$. According to Peto and Peto (1972), this weight is more robust for cases where the censoring between the two groups may follow different patterns.
5. Modified Peto-Peto: $W(t) = \frac{\bar{S}(t_i)Y_i}{(Y_i+1)}$. Andersen et. al. (1982) proposed this slight modification to the Peto-Peto weight.
6. Fleming-Harrington: $W(t) = \bar{S}(t_{i-1})^p [1 - \bar{S}(t_{i-1})]^q$. This is a family of weights determined by the powers p and q . Setting p and q to 0 gives the Log-rank test. Setting p to 1 and q to 0 gives a weight very similar to the Peto-Peto weight. It can be seen that a large p relative to q will represent a weighting favoring early events whereas a $q > p$ favors later events (Fleming and Harrington, 1981).

4.3 Brownian Motion

In order to understand the Renyi family of statistics, one needs to know a little bit about Brownian Motion. Brownian Motion is a special type of Markov chain; some would call it one of the simplest types of Markov chains. In addition to satisfying the criterion to be a Markov chain, the following properties also hold for a model satisfying Brownian Motion:

1. For all time $h > 0$, the difference $X(t+h) - X(t)$ has a normal distribution
2. The difference $X(t+h) - X(t)$, $0 < t_1 < t_2 < \dots < t_n$ are mutually independent
3. The mean difference is zero

Following these principles, it becomes clear that standard Brownian Motion is intended to describe random noise that oscillates randomly and follows a Gaussian distribution.

4.4 Renyi Mathematics

The Renyi test statistic is a direct extension of the above calculations. Instead of taking the sum over all of the time points, the numerator of the Renyi statistic is simply $M = \max(Z_k(t))$ for a single time point t . The corresponding test

statistic is $B(t) = \frac{M}{\sqrt{\sigma_{kk}}}$, where the distribution of B can be approximated by a standard Brownian motion process. This p-value can be computed as

$$Pr[\max|B(t)| > y] = 1 - \frac{4}{\pi} \sum_{k=0}^{\infty} \frac{(-1)^k}{2k+1} e^{-\frac{\pi^2(2k+1)^2}{8y^2}}$$

Consider for a moment the implications that this motion has on survival analysis. When the Renyi family of statistics is used to model the true difference in survival rates between the two groups, it is assuming that the process of events occurring follows a random mean-zero Gaussian motion process, implying that the groups should be no different as any difference in their survival rates would only be due to chance. Therefore, when we take the largest difference between the survival times of the two groups, we are in essence trying to categorize the random motion in one difference by claiming that if a single difference is too large, these curves cannot be different due solely to random noise and there must be a true underlying difference in the survival curves due to the attributes of the strata being compared. When we consider that survival analysis as a whole is based off of the Martingale family of stochastic processes, it comes as no surprise that the difference in survival between the two groups can be accurately measured by another type of stochastic process.

5. CALCULATING THE RENYI STATISTICS IN SAS


For the benefit of the reader, the macro %_renyi has been included in Appendix A of this paper. While this macro is designed for general use, the following is an outline of how the macro was developed:

- First, develop the life table used to determine the Kaplan-Meier statistics complete with censored data, failure data, estimated survival times and at risk sets.
- Next, derive the classical log-rank test. As a quality control measure, ensure that the log-rank statistic produced matches exactly what comes out of a classical Kaplan-Meier analysis using PROC LIFETEST. The appropriate weight should be applied in this step.
- Third, find the time point and corresponding statistic whose absolute value contributes the largest to the log-rank test statistic.
- To calculate the Renyi statistic, take the value in the last step and divide by the standard deviation of the classical log-rank test to obtain the Renyi test statistic.
- Use the formula supplied by Billingsly (1968) to calculate the associated p-value (note that determining infinite sums in SAS is difficult, so the sum from 0 to 10,000 was used as an approximate p-value).

6. A DATA ANALYSIS EXAMPLE USING RENYI STATISTICS


Consider the following example presented in *Survival Analysis: Techniques for Censored and Truncated Data* (Klein, 2003).

A clinical trial of chemotherapy against chemotherapy combined with radiotherapy in the treatment of locally unresectable gastric cancer was conducted by the Gastrointestinal Tumor Study Group (1982). In this trial, forty-five patients were randomized to each of the two arms and followed for about eight years.



Graphically, there is good evidence that we have a case of crossing hazards. First, when the survival curves cross, there is a strong indication that the rate of failure between the two groups also crosses at some point. Secondly, while PROC LIFETEST will not provide estimates of the hazard functions when using the Kaplan-Meier method, it will provide hazard estimates using the life table method. Above is the output for the estimated hazard functions from the life table method. Clearly, using those estimations, the hazard rates between the two cross.

In addition, the plot of $\log(-\log(\text{Survival}))$ against $\log(\text{time})$ should result in parallel curves if proportional hazards exist. However, with these data, that curve agrees with the others that the hazard rates in the two groups are not proportional.



All these evidence suggests that crossing hazard rates may exist. Thus the Renyi statistics need to be considered. Empirically, it appears that there is a difference in the survival rates between the two groups, although which group is better is difficult to ascertain. The chemotherapy only group appears to have a better survival rate until roughly day 1000 when the chemotherapy plus radiation survival rate surpasses the chemotherapy only group and remains higher until the end of the study. Remember that these statistics are designed only to explain if there is a difference between the two groups, not necessarily that one group is better than another.

Remember that the classical log-rank test is a sum of estimated difference over all applicable time points. Thus, in this trial, the signs of the estimated difference will switch as the hazard rates cross (i.e. they may all be negative before the cross of the survival curves and positive after or vice versa). Using the LIFETEST procedure in SAS, we observe a chi-square test-statistic of 0.2319 with one degree of freedom (p-value = 0.6301), indicating no significant difference between the two groups. As expected, the crossing of the survival curves caused the test statistic to cancel itself out rendering a verdict that may not truly represent the observed data. The Wilcoxon test may prove to be more significant as it places a higher weight on the earlier observations. Given the fact that the cross in the survival rates does not happen until roughly day 1000, this test may prove to be more significant. Using PROC LIFETEST, we obtain a chi-square Wilcoxon test statistic of 3.997 with one degree of freedom (p-value = 0.0456), significant at the .05 level. It is disconcerting that there is such a disparity between the two tests with one test showing significance and the other showing a complete lack of difference between the two groups.

Let us look also to the Cox Proportional Hazards model for help. Consider the following (simple) model:

```
PROC PHREG;
MODEL TIME*CENSOR(0)=GROUP;
RUN;
```

Running this model, we obtain a chi-square test statistic for the significance of the group effect of .2315 with one degree of freedom (p-value = 0.6304). This is highly non-significant at the .05 level indicating that there is no significant difference between then two groups at the .05 level (notice how close this test statistic is to the classical log-rank test: p-value of .6301 for the log-rank test and .6304 for the Cox Proportional Hazards model).

Now, consider the Renyi family of statistics. Remember that all of the weights for the classical log-rank test can be used for the Renyi family of statistics. We will consider the two side by side in the table below:

Weight	Classical Log-Rank P-value	Renyi Family: P-value	Renyi Family: Time
Log-Rank	0.630	0.056	315
Gehan (Wilcoxon)	0.046	0.006	315
Tarone-Ware	0.165	0.015	315
Peto-Peto	0.045	0.006	315
Modified Peto-Peto	0.042	0.006	315
Fleming Harrington (p=0, q=1)	0.153	0.305	2363
Fleming Harrington (p=1, q=0)	0.046	0.006	315
Fleming Harrington (p=1, q=1)	0.916	0.236	315

Notice the range of p-values in the classical testing case (0.0424 – 0.9160). Depending on the weight chosen, the testing will have dramatically different effects. When the Renyi family is used, this range decreased dramatically (0.006 – 0.305), indicating that the Renyi family more accurately describes the true difference between the two groups. The log-rank test failed to detect the hazard rate difference due to its low power of detecting crossing hazards while the Renyi statistics are able to show differences for the majority of the weights.

Secondly, notice the times in the third column. Remember that the Renyi statistic is built around the largest difference in the test statistic between the two groups at any time over the curves. Observing the original survival curves,

it is no surprise that most of the tests find the largest difference between the two groups at time 315 since, empirically, this time contains the largest difference between the two curves. However, the Fleming Harrington (0,1) test finds its largest difference at time 2363. Remember that, in the Fleming and Harrington tests, a large value of p compared to q emphasizes differences early in the curves where the opposite emphasizes differences later in the curves. In this case, since q is 1 and p is 0, the differences later in the curves are emphasized, leading to the later time point of 2363 as the largest difference in test statistic between the two groups.

Lastly, the interpretation of a significant Renyi test statistic is not that the survival rate in the chemotherapy only group is significantly greater than the chemotherapy plus radiation group. It does mean that there is a significant difference in the survival rates between the two groups, but does not imply that one group is superior to another. In this case, it is clear that the chemotherapy only group has a higher rate of survival in the beginning of study compared to the chemotherapy plus radiation but a worse survival rate afterwards. Therefore, careful interpretation of a significant Renyi test statistic is required.

7. CONCLUSION AND DISCUSSION

As evidenced above, crossing hazard rates can be detrimental to the analysis of survival data using classical Log-Rank/Kaplan-Meier or Cox Proportional Hazards methods. With the use of the Renyi family of test statistics, the data can be analyzed appropriately without the use and interpretation of time-dependent covariates to accurately describe the difference between the survival rates of the two curves. The various options associated with the Renyi statistic allow for application to many situations where crossing hazards occur.

An alternative way to analyze survival data when hazards cross is to use time-dependent covariate in the Cox model. This approach is more complex computationally but allows direct estimates of the hazard ratio and covariate effects. Renyi statistics are easy to compute but does not generate the actual estimate of the group effects. In summary, Renyi statistics provide an alternative to time-dependent Cox regression model in analyzing data when proportional hazard assumptions are violated.

8. REFERENCES

- Andersen, P.K. Testing Goodness of Fit of Cox's Regression and Life Model. *Biometrics* 38 (1982): 67-77.
- Billingsly, P. *Convergence of Probability Measures*. New York: John Wiley and Sons, 1968.
- Fleming, T.R. and Harrington, D.P. *Counting Processes and Survival Analysis*. New York: John Wiley and Sons, 1991.
- Gehan, E. A. A Generalized Wilcoxon Test for Comparing Arbitrarily Singly Censored Samples. *Biometrika* 52 (1965): 203-223.
- Gill, R.D. Censoring and Stochastic Integrals. *Mathematical Centre Tracts*. Amsterdam: Mathematisch Centrum, 1980, 124.
- Klein, John P. and Moeschberger, Melvin L. *Survival Analysis: Techniques for Censored and Truncated Data, Second Edition*. New York: Springer, 2003.
- Peto, R. and Peto, J. Asymptotically Efficient Rank Invariant Test Procedures (with Discussion). *Journal of the Royal Statistical Society A* 135 (1972): 185-206.
- Rényi, Alfréd. On the Theory of Order Statistics. *Acta Mathematica Hungarica* 4 (1953): 191-231.
- Schumacher, M. Two-Sample Tests of Cramér-Von Mises and Kolmogorov-Smirnov Type for Randomly Censored Data. *International Statistical Review* 52 (1984): 263-281.

APPENDIX I: %_Renyi Macro

```

*****
* Macro: _RENYI
* Author: Mat Davis
* Affiliation: University of Pennsylvania
* Email: davismat@mail.med.upenn.edu
* Description: This macro will obtain the Renyi Test Statistic
*              testing the null hypothesis that the hazard rates
*              for two stratified groups of subjects are equal.
*****
*              MACRO PARAMETERS
*
* DATA: The dataset to be analyzed
* STRATA: The variable that indicates the group to be stratified
* TIME: The variable that indicates survival time
* CENSOR: The variable that indicates censored status
* CVAL: The value of the censored observations in CENSOR
* WEIGHT: The desired weight for the analysis
* LOGRANK: 1
* GEHAN: Combined at risk set
* TARONEWARE: The square root of the
*              combined risk set at time t
* PETOPETO: The combined survival rate at
*              time t
* MODPETOPETO: Corrected survival rate at
*              time t
* FH: (s(t*))^p*(1-s(t*))^q where s(t) is the
*              combined survival rate at time t-1
* FH1: The value of 'p' for the Fleming-Harrington weights
* FH2: The value of 'q' for the Fleming-Harrington weights
*****

%macro _renyi(data,strata,time,censor,cval,weight,fh1,fh2);

%macro _fillgap(old,new);
if not missing(&old.) then &new.=&old.;
%mend;

proc lifetest method=km plots=(s) data=&data.;
strata &strata.;
time &time.*&censor.(&cval.);
ods output productlimitestimates=prod;
run;

proc means data=prod noprint;
class &time. &strata.;
var failed;
output out=prodmax(where=(not missing(&time.) and not missing(&strata.))) max=failed;
run;quit;

proc means data=prod noprint;
class &time. &strata.;
var censor;
output out=cenmax(where=(not missing(&time.) and not missing(&strata.))) max=censor;
run;quit;

proc means data=prod noprint;
class &time. &strata.;
var left;
output out=prodmin(where=(not missing(&time.) and not missing(&strata.))) min=left;
run;quit;

data prod;
merge prodmax(keep=&time. &strata. failed) prodmin(keep=&time. &strata. left) cenmax(keep=&time. &strata.
censor);
by &time. &strata.;
run;

```

```

data prod1 prod2;
set prod;
run;

data prodx(keep=&time. d1 d2 y1 y2 totf renyi1 censor1 censor2 renyisum
absrenyisum renyisd renyisdsum absrenyisdsum _id atrisk atrisk1 atrisk2
lastd1 lastd2 lastcensor1 lastcensor2 survival weight lastsurv);
retain g1 g2 f1 f2 l1 l2 renyisum 0 renyisdsum 0 atrisk atrisk1 atrisk2 survival 1;
merge prod1(where=(&strata.1=1) rename=(&strata.=&strata.1 failed=failed1 left=left1 censor=censor1))
      prod2(where=(&strata.2=2) rename=(&strata.=&strata.2 failed=failed2 left=left2 censor=censor2));
by &time.;

***DATA MANIPULATION TO OBTAIN FULL TABLE***;

%_fillgap(&strata.1,g1) %_fillgap(&strata.2,g2) %_fillgap(failed1,f1)
%_fillgap(failed2,f2) %_fillgap(left1,l1) %_fillgap(left2,l2)
if missing(censor1) then censor1=0; if missing(censor2) then censor2=0;
lastf1=lag(f1); lastf2=lag(f2); lastl1=lag(l1); lastl2=lag(l2);
lastcensor1=lag(censor1); lastcensor2=lag(censor2);
d1=f1-lastf1; d2=f2-lastf2;
y1=l1+d1+censor1; y2=l2+d2+censor2;
lastd1=lag(d1); lastd2=lag(d2); totf=sum(d1,d2);

***REMOVE ENTRY AT TIME=0***;

if &time.=0 then delete;

***GET THE AT RISK SET***;

if _N_=2 then atrisk=sum(y1,y2);
else atrisk=atrisk-lastd1-lastd2-lastcensor1-lastcensor2;
if _N_=2 then atrisk1=y1;
else atrisk1=atrisk1-lastd1-lastcensor1;
if _N_=2 then atrisk2=y2;
else atrisk2=atrisk2-lastd2-lastcensor2;

***DETERMINE SURVIVAL RATES***;

survpart=(1-(totf/(atrisk+1)));
if not missing(survpart) then survival=survival*survpart;
lastsurv=survival/survpart;

***Correct for 0^0***;

if lastsurv=0 then lastsurv=.0000001;
if lastsurv=1 then lastsurv=.9999999;

***SELECTION OF THE WEIGHT***;

%if %upcase(&weight.)=LOGRANK %then %do;

weight=1;

%end;

%else %if %upcase(&weight.)=GEHAN %then %do;

weight=atrisk;

%end;

%else %if %upcase(&weight.)=TARONEWARE %then %do;

weight=atrisk**.5;

%end;

%else %if %upcase(&weight.)=PETOPETO %then %do;

weight=survival;

```

```

%end;

%else %if %upcase(&weight.)=MODPETOPETO %then %do;

weight=survival*atrisk/(atrisk+1);

%end;

%else %if %upcase(&weight.)=FH %then %do;

weight=lastsurv**(&fh1.)*(1-lastsurv)**(&fh2.);

%end;

***OBTAIN THE RENYI STATISTICS***;

if atrisk=0 then renyi1=0; else renyi1=weight*(d1-atrisk1*(toff/atrisk));
renyisum=renyisum+renyi1;
absrenyisum=abs(renyisum);
if atrisk-1=0 then renyisd=0; else renyisd=weight**2*(atrisk1/atrisk)*(atrisk2/atrisk)*((atrisk-toff)/(atrisk-1))*toff;
renyisdsdsum=renyisd+renyisdsdsum;
absrenyisdsdsum=abs(renyisdsdsum);
_id=1;
run;quit;

***OBTAIN THE SUM OF THE RENYI STATISTICS (TEST STATISTIC)***;

proc means data=prodx sum noprint;
var renyi1;
output out=lr sum=lr;
run;quit;

data _null_;
set lr;
call symput('lr',lr);
run;

proc sort data=prodx;
by &time.;
run;

***GET RENYI SD***;

data _null_;
set prodx;
by _id;
if last._id;
call symput('renyisdsdsum',renyisdsdsum);
run;

proc sort data=prodx;
by descending absrenyisum &time.;
run;

***REPORT RENYI STATISTICS***;

data renyi;
set prodx;
by descending absrenyisum &time.;
if _N_=1;
renyisdeval=sqrt(&renyisdsdsum.);
renyistat=absrenyisum/sqrt(&renyisdsdsum.);
file print;
put *****.
put *****RENYI STATISTICS*****.
put *****.
put " ";
put "Time that maximizes Z(t): " &time.;

```

```

put "%upcase(&weight.) statistic (numerator): &lr.";
put "Z(t) Statistic: " absrenyisum;
put "Z(t) Standard Deviation: " renyisdeval ;
put "Renyi Test Statistic: " renyistat;
***Initialize Infsum***;
infsum=0;
%do k=0 %to 10000;
infsum=infsum+(-1)**%eval(&k.)/(2**%eval(&k.)+1)*exp(-1*constant('PI')**2*(2**%eval(&k.)+1)**2/(8*renyistat**2));
%end;
pvalue=1-(4/constant('PI'))*infsum;
put " Approximate P-value: " pvalue;
put " ";
put "*****";
run;

%mend;

```

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Mat Davis
507 Blockley Hall
423 Guardian Drive, Philadelphia, PA 19104
davismat@mail.med.upenn.edu

Sharon Xie
618 Blockley Hall
423 Guardian Drive, Philadelphia, PA 19104
sxie@mail.med.upenn.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.
Other brand and product names are trademarks of their respective companies.