

Generate Informative Clinical Laboratory Results Listing

Sai Ma, Everest Clinical Research Services Inc., Markham, Ontario Canada

ABSTRACT

Clinical laboratory results review is an indispensable, or even crucial, part of clinical trial studies. An informative, neat-looking laboratory results listing can provide insights on important clinical findings or trend identification. The nature of laboratory data (multi-visit, massive size, and often messy) sometimes poses a challenge to obtain such a listing. A method utilizing SAS® procedure PROC REPORT is presented in this paper to achieve this.

KEY WORDS

PROC REPORT, Laboratory, Listing

INTRODUCTION

In the study at hand, a listing of laboratory data patient profile is desired for clinical review (Figure 1).

Listing: Clinical Laboratory Results (All Randomized)

Patient ID: xxxxxxxx		Study Center: xxx		Sex: x	Age: xx	Race: xxxxx	Treatment Group: xxxxx	
Date of First - Last Dose of Double-Blind Study Medication:						yyyy-mm-dd - yyyy-mm-dd		
Laboratory Test	Unit	Normal Range		Assessment Date (CRF, Derived Visit)				
		Low	High	yyyy-mm-dd (Visit1, Screening)	yyyy-mm-dd (Visit4, Baseline)	yyyy-mm-dd (Visit14, Week 26)	
Hematology								
Basophils	%	x	x	x	x	x	x	x
.....								

Figure 1. Listing Shell

At first glance, the following programming issues must be tackled in a dynamic way:

- Display subject information in the listing header
- Use the unique test dates/visits (both scheduled and unscheduled) as the column headers
- Show as many dates/visits a subject could have during the study

The compute block in PROC REPORT is an ideal tool to resolve the first issue. The second issue can be accomplished by defining an array of macro variables. The last issue is more involved. The difficulty is two-fold: First, since each subject can have more than 10 scheduled/unscheduled visits with unique test dates, fitting all data on a single page for every subject is not feasible. After balancing the amount of information each page accommodates and the page space utilization, only a limited number of dates/visits (4, for this study) can be contained in one page for the same laboratory test. The rest will continuously span over the next page(s). For ease of review, the relevant laboratory test information (Laboratory Test Group, Laboratory Test Name, Unit, Low and High Normal Range) also needs to be repeated in this case. Second, in a perfect "Lab world", every subject should have the same number of visits and take the same group of laboratory tests per protocol. Unfortunately, this almost never happens in our real-life programming practice. Number of visits can vary dramatically for different subjects and/or laboratory test groups (Hematology, Chemistry, Urinalysis, Urine Drug Screen). This implies each test group should be evaluated separately so that the collected data will expand over the exact number of pages as needed.

DATA PREPARATION

As the first step to address the above mentioned issues, a variable of date sequence (DTSEQ) and a variable of the total number of unique dates (MXDTSEQ) are created after sorting the data by the unique dates and visits within test group and subject. This is the key to dynamically displaying dates/visits in the proper order. Subsequently, all the desired information is defined into a group of macro variables to further subset or loop data in the reporting step.

```

data final;
  set final0;
  length dtnames $100;
  by subnum __labgn dtseq;
  if first.subnum then do;
    sidno+1;
    labgn=0;
  end;
  if first.__labgn then labgn+1;
  dtnames='d' || trim(left(put(sidno, best.))) || '_' ||
    trim(left(put(labgn, best.))) || trim(left(put(dtseq, best.)));
  mxlabdts='mdt' || trim(left(put(sidno, best.))) || '_' ||
    trim(left(put(labgn, best.)));
  mxgrp='mxgp_' || trim(left(put(sidno, best.)));
  sids='sid' || trim(left(put(sidno, best.)));
  labgrps='grp' || trim(left(put(sidno, best.))) || '_' || trim(left(put(labgn, best.)));
  if last.dtseq then call symput(dtnames, trim(left(dates)));
  if last.__labgn then do;
    call symput(mxlabdts, trim(left(put(mxdtseq, best.))));
    call symput(labgrps, trim(left(put(__labgn, best.))));
  end;
  if last.subnum then do;
    call symput(sids, trim(left(subnum)));
    call symput(mxgrp, trim(left(put(labgn, best.))));
  end;
run;

```

Sample data for Chemistry laboratory parameters (test group numeric value is 9, defined by the internal lab database, but 7 as collected) of the first subject (total 11 unique dates/visits) and macro variable assignments are given in Figure 2.

SUBNUM	__LABGP	__LABGN	DTSEQ	DATES
9999999	Chemistry	9	1	2999-04-19!(Visit 1,!Screening!/!Baseline)
9999999	Chemistry	9	2	2999-05-13!(Visit 4,!Baseline)
9999999	Chemistry	9	3	2999-05-31!(Visit 6,!Week 2)
9999999	Chemistry	9	4	2999-06-29!(Visit 8,!Week 6)
9999999	Chemistry	9	5	2999-07-27!(Visit 10,!Week 10)
9999999	Chemistry	9	6	2999-08-23!(Visit 11,!Week 14)
9999999	Chemistry	9	7	2999-09-22!(Visit 12,!Week 18)
9999999	Chemistry	9	8	2999-10-14!(UNS,)
9999999	Chemistry	9	9	2999-10-19!(Visit 13,!Week 22!/!End of DBT!/!End of Study)
9999999	Chemistry	9	10	2999-11-09!(UNS,)
9999999	Chemistry	9	11	2999-11-16!(Visit 14,!Week 26!/!End of DBT!/!End of Study)

SUBNUM	DTNAMES	SIDNO	LABGN	MXLABDTS	MXGRP	SIDS	LABGRPS
9999999	d1_71	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_72	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_73	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_74	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_75	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_76	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_77	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_78	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_79	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_710	1	7	mdt1_7	mxgp_1	sid1	grp1_7
9999999	d1_711	1	7	mdt1_7	mxgp_1	sid1	grp1_7

Figure 2. Sample Outputs - Subject ID 9999999 Chemistry Parameters

The last bit of data manipulation involves transposing the dataset by the group of variables that will be repeated across pages. Variable DTSEQ is used in the ID statement of PROC TRANSPOSE. A snapshot of the dataset is in Figure 3. Also, a macro variable of the total number of randomized subjects is defined (&N_RAND), and all the desired subject information is concatenated into two title variables, TITL1 and TITL2.

__NAME	LAB_UNT	LAB_LW	LAB_HG	DT_1	DT_2	DT_10	DT_11
Adiponectin, Total	ug/mL	1.111	99.999		5.555		5.555
Alanine Aminotransferase (SGPT)	U/L	1	99	55	55	55	55
Albumin	G/L	11	99	55	55	55	55
Alkaline Phosphatase	U/L	11	999	55	55	55	55
Aspartate Aminotransferase (SGOT)	U/L	1	99	55	55	55	55
Bilirubin, Total	UMOL/L	1	99	5	5	5	5
Blood Urea Nitrogen	MMOL/L	1.1	9.9	5.5	5.5	5.5	5.5
Calcium	MMOL/L	1.11	9.99	5.55	5.55	5.55 L	5.55 L
Chloride	MMOL/L	11	999			55	55
Cholesterol, HDL	mmol/L	1.11	9.99	5.55			5.55
Cholesterol, LDL Calculated	mmol/L	1.11	9.99	5.55			5.55 L
Cholesterol, Total	mmol/L	1.11	9.99	5.55 L			5.55 L
Creatinine	UMOL/L	11	99	55	55	55	55
Fructosamine	umol/L	111	999		555		555 L
GFR Extended Calculation	ML/MIN/SSA	CCCCCCCC.	CCCCC	555			555
Glucose, Fasting	MMOL/L	1.1	9.9		5.5 EL		
Glucose, Plasma, Fasting	mmol/L	1.1	9.9		5.5 L		55.5 EL
Insulin	mIU/L	1.1	99.9		55.5 L		55.5 L
Potassium	MMOL/L	1.1	9.9	5.5	5.5	5.5	5.5
Protein, Total	G/L	11	99			55	55
Sodium	MMOL/L	111	999	555	555	555	555
Thyroid Stimulating Hormone	MIU/L	1.111	9.999		5.555		
Triglycerides, Fasting	mmol/L	1.11	9.99	5.55 EL			5.55 EL

Figure 3. Sample Outputs - Transposed Data

DATA REPORTING

The final output is generated by PROC REPORT (Figure 4). A WHERE statement guarantees each parameter group for different subjects will stretch over the minimum number of pages for available dates/visits. The ID option in the DEFINE statement ensures the selected columns repeat for each page for the same laboratory test, if, the total number of unique dates/visits exceeds 4. A text string "(Continuing...)" is added from the second page onward to indicate the page continuation and to enhance the readability. The value of PROC REPORT option COLWIDTH (width of variable BLKSPACE) is controlled to pad blank spaces so that every page has the desired 132 spaces for length. As discussed earlier, compute block is used to add subject information in the header portion. The full piece of PROC REPORT programs is provided in the Appendix.

AAAAA BBBBBBBB CCCCC								ZZZZZZZZ
Protocol XXXXXX001								Page xx1 of yyy
Listing: Clinical Laboratory Results (All Randomized)								
Patient ID:9999999 Study Center:999 Sex:X Age:00 Race:00000 Treatment Group:XOXO								
Date of First - Last Dose of Double-Blind Study Medication: 2999-05-17 - 2999-11-28								
Assessment Date (CRF, Derived Visit)								
Laboratory Test	Unit	Normal Range		2999-04-19 (Visit 1, Screening/ Baseline)	2999-05-13 (Visit 4, Baseline)	2999-05-31 (Visit 6, Week 2)	2999-06-29 (Visit 8, Week 6)	
		Low	High					
Chemistry								
Adiponectin, Total	ug/mL	1.111	99.999		5.555			
Alanine Aminotransferase (SGPT)	U/L	1	99	55	55			
Albumin	G/L	11	99	55	55			
Alkaline Phosphatase	U/L	11	999	55	55			
Aspartate Aminotransferase (SGOT)	U/L	1	99	55	55			
Bilirubin, Total	UMOL/L	1	99	5	5			
Blood Urea Nitrogen	MMOL/L	1.1	9.9	5.5	5.5	5.5		5.5
Calcium	MMOL/L	1.11	9.99	5.55	5.55			
Chloride	MMOL/L	11	999			555		55
.....								

Listing: Clinical Laboratory Results (All Randomized)

Patient ID:9999999 Study Center:999 Sex:X Age:00 Race:O0000 Treatment Group:XOXO

Date of First - Last Dose of Double-Blind Study Medication: 2999-05-17 - 2999-11-28

Laboratory Test	Unit	Assessment Date (CRF, Derived Visit)					
		(Continuing...)					
		Normal Range		2999-07-27 (Visit 10, Week 10)	2999-08-23 (Visit 11, Week 14)	2999-09-22 (Visit 12, Week 18)	2999-10-14 (UNS,)
Low	High						
Chemistry							
Adiponectin, Total	ug/mL	1.111	99.999				
Alanine Aminotransferase (SGPT)	U/L	1	99	55			55
Albumin	G/L	11	99	55			55
Alkaline Phosphatase	U/L	11	999	55			55
Aspartate Aminotransferase (SGOT)	U/L	1	99	55			55
Bilirubin, Total	UMOL/L	1	99	5			5
Blood Urea Nitrogen	MMOL/L	1.1	9.9	5.5	5.5	5.5	5.5
Calcium	MMOL/L	1.11	9.99	5.55			5.55
Chloride	MMOL/L	11	999		555	55	
.....							

Listing: Clinical Laboratory Results (All Randomized)

Patient ID:9999999 Study Center:999 Sex:X Age:00 Race:O0000 Treatment Group:XOXO

Date of First - Last Dose of Double-Blind Study Medication: 2000-05-17 - 2000-11-28

Laboratory Test	Unit	Assessment Date (CRF, Derived Visit)				
		(Continuing...)				
		Normal Range		2999-10-19 (Visit 13, Week 22/ End of DBT/ End of Study)	2999-11-09 (UNS,)	2999-11-16 (Visit 14, Week 26/ End of DBT/ End of Study)
Low	High					
Chemistry						
Adiponectin, Total	ug/mL	1.111	99.999			5.555
Alanine Aminotransferase (SGPT)	U/L	1	99		55	55
Albumin	G/L	11	99		55	55
Alkaline Phosphatase	U/L	11	999		55	55
Aspartate Aminotransferase (SGOT)	U/L	1	99		55	55
Bilirubin, Total	UMOL/L	1	99		5	5
Blood Urea Nitrogen	MMOL/L	1.1	9.9	5.5	5.5	5.5
Calcium	MMOL/L	1.11	9.99		5.55 L	5.55 L
Chloride	MMOL/L	11	999	555	55	55
.....						

Figure 4. Sample Outputs - Final Layout

CONCLUSION

Depending on what and how you want to display your data, dynamic listing/table with similar structure (e.g. patient profile for other panels) can easily adopt the idea from this paper. PROC REPORT is powerful. Take your time and play with it, you can usually get what you want, no matter how fancy your output should look. If time is a luxury you cannot afford, hopefully, this paper can give you a start.

REFERENCES

Carpenter, Arthur L. (2008), "PROC REPORT: Compute Block Basics - Part I Tutorial", SAS Global Forum 2008 Proceedings, San Antonio, Texas, March 16-19, 2008.

ACKNOWLEDGMENTS

The author wants to thank Suwen Li and Tania George for their invaluable comments on this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Sai Ma
Enterprise: Everest Clinical Research Services Inc.
Address: 675 Cochrane Drive, Suite 408, East Tower
City, State ZIP: Markham, Ontario Canada L3R 0B8
Work Phone: (905) 752-5238
Fax: (905) 752-5224
E-mail: sai.ma@ecrscorp.com
Web: www.ecrscorp.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX - PROC REPORT PROGRAM

```
%do zz=1 %to &n_rand;
  %do yy=1 %to &&mxgp_&zz;
    proc report data=final nowd missing headline headskip spacing=0 split='!'
      colwidth=%if &&mdt&zz._&yy gt 4 and %sysfunc(mod(&&mdt&zz._&yy, 4)) ne 0
        %then %eval((4-%sysfunc(mod(&&mdt&zz._&yy, 4)))*15);
        %else %if &&mdt&zz._&yy gt 4 and
          %sysfunc(mod(&&mdt&zz._&yy, 4)) eq 0 %then 0;
        %else %if &&mdt&zz._&yy le 4 %then %eval((4-&&mdt&zz._&yy)*15);
      ;
    column (titl1 titl2 __labgp __labgn __name lab_unt
      ('Normal Range!__!' lab_lw lab_hg)
      ('Assessment Date!(CRF, Derived Visit)!__!'
        %if &&mdt&zz._&yy le 4 %then dt_1-dt_&&mdt&zz._&yy blkspace;
        %else dt_1-dt_4 ('(Continuing...)' dt_5-dt_&&mdt&zz._&yy blkspace)););
    where subnum eq "&&sid&zz" and __labgn eq &&grp&zz._&yy;
    define titl1 /group order=data noprint;
    define titl2 /group order=data noprint;
    define __labgp /group order=data noprint;
    define __labgn /group order=internal noprint;
    define __name /order order=internal width=36 flow left id "Laboratory Test";
    define lab_unt /order order=internal width=10 flow left id "Unit";
    define lab_lw /order order=internal width=11 flow left id "Low" spacing=3;
    define lab_hg /order order=internal width=11 flow left id "High" spacing=1;
    %do xx=1 %to &&mdt&zz._&yy;
      define dt_&xx /order order=internal width=15 flow left "&&d&zz._&yy.&xx";
    %end;
    define blkspace / " ";
    compute before __labgn;
      line @01 __labgp $100.;
    endcomp;
    compute before _page_;
      line 132*'_';
      line @1 titl1 $130.;
      line @1 " ";
      line @1 titl2 $130.;
    endcomp;
  run;
%end;
%end;
```